FORTRAN
应用程序库

上海机械学院
安徽省计算中心 编写

FORTRAN 应用程序库

上海机械学院
安徽省计算中心 编写

上海科学技术文献出版社

FORTRAN应用程序库

上海机械学院
安徽省计算中心
编写

*

上海科学技术文献出版社出版
(上海市武康路2号)
新华书店上海发行所发行
上海市印刷十二厂印刷

*

开本787×1092 1/32 印张9.5 字数220,000
1984年8月第1版 1984年8月第4次印刷
印数: 1—38,500

书号: 15192 317 定价: 1.17元

《科技新书目》75-200

序　　言

随着我国国民经济和科学技术的发展，电子计算机的使用日益广泛。科技计算、企业管理和数据统计分析是电子计算机应用的几个重要方面。为了普及与推广计算机的应用，提高计算机的使用率，缩短解题周期，尤其是编写和调试程序的周期，我们编写了 FORTRAN 应用程序库。

FORTRAN 语言是当今世界上最流行的高级程序设计语言之一，它特别适合解决各类计算问题。凡配有通用的 FORTRAN IV 的各种计算机都可以使用本书的应用程序。

本书共有 54 个程序，分为九章，依次为插值与逼近(6 个程序)、方程求根(3 个)、数值积分(4 个)、线性代数计算(8 个)、常微分方程的数值解法(6 个)、图论(5 个)、最优化方法(5 个)、多元统计分析(7 个)、线性规划(10 个)。

每个程序包括功能、方法简介、使用说明、程序与例题等五个部分。全部程序不但在 CJ-709 机上调试通过，而且在 CROMEMCO 微型计算机上进行了验算。本书参考 CROMEMCO 微机有关输入、输出的规定，其设备号取为 5，对于其它型号的计算机，仅在这一点上可能不同。

参加本书编写工作的主要人员有上海机械学院王家源、朱自强、林建浩同志；周静芳、王仰一同志也参加了编写工作；安徽省计算中心笪祖勤、程少春、颜锦纯、陈琪霞、李本超同志编写了第一章中第五、六两节及第八章。上海科技大学黄育仁、董炳华、叶秀明、张荣欣等同志通读了原稿，并提出了许多有益的意见，

在编写过程中黄育仁同志和复旦大学许自省同志给予热情指导，在此表示衷心感谢。

由于我们水平有限，文中错误与不妥之处，请读者批评指正。

编　者

一九八三年九月

目　　录

第一章　插值与逼近

1.1　拉格朗日插值

一、功能

用拉格朗日插值公式对给定的 n 个插值结点进行插值。

二、方法简介

对给定的 n 个插值结点 $x_1, x_2, \cdots, x_n$ 及对应的函数值 $y_1, y_2, \cdots, y_n$，应用拉格朗日插值公式，计算在 x 点处的函数值 $y(x)$：

$$y(x)=\sum_{k=1}^{n}\left(\prod_{\substack{j=1\\ j\neq k}}^{n}\frac{x-x_j}{x_k-x_j}\right)y_k$$

三、使用说明

1．子程序语句

SUBROUTINE　LAGR(X0, Y0, N, X, Y)

2．哑元说明

输入参数：

N　整变量，插值结点个数。

X0　N 个元素的一维实数组，存放给定的插值结点。

Y0　N 个元素的一维实数组，存放插值结点上相应的函数值。

X　实变量，插值点。

输出参数：

Y　实变量，插值结果。

四、程序

```
SUBROUTINE LAGR(X0, Y0, N, X, Y)
DIMENSION X0(N), Y0(N)
```

```
      Y=0.0
      DO 30 I=1, N
      P=1.0
      DO 20 J=1, N
      IF (I-J) 10, 20, 10
10    P=P*(X-X0(J))/(X0(I)-X0(J))
20    CONTINUE
30    Y=Y+P*Y0(I)
      RETURN
      END
```

五、例题

已知如表 1-1 所示的列表函数，求 $x=0.472$ 处的函数值。

表 1-1

x	0.46	0.47	0.48	0.49
y	0.484655	0.493745	0.502750	0.511668

用数组 X0 的元素顺序存放 0.46, 0.47, 0.48, 0.49; 数组 Y0 的元素顺序存放 0.484655, 0.493745, 0.502750, 0.511668。

程序如下：

```
      SUBROUTINE  LAGR(X0, Y0, N, X, Y)
      ⋮本子程序段体部分
      END

      DIMENSION X0(4), Y0(4)
      READ(5, 1)X0, Y0
1     FORMAT(4F8.3/4F10.6)
      X=0.472
      CALL LAGR(X0, Y0, 4, X, Y)
      WRITE(5, 20)X, Y
20    FORMAT(4X, 'X=', F6.3, 6X, 'Y=', F8.5)
      STOP
      END
```

计算结果：　　$y(0.472)=0.49555$

1.2 分段抛物插值

一、功能

用拉格朗日三点插值公式对一元插值表中选取最靠近插值点 x 的相邻三点进行插值。

二、方法简介

设给定的 n 个插值结点 $x_1<x_2<\cdots<x_n$ 及对应的函数值 $y_1, y_2, \cdots, y_n$, 取三点 x_{i-1}, x_i 和 x_{i+1} 按下列公式进行插值:

$$
\begin{aligned}
y= & \frac{(x-x_i)(x-x_{i+1})}{(x_{i-1}-x_i)(x_{i-1}-x_{i+1})} y_{i-1} \\
& +\frac{(x-x_{i-1})(x-x_{i+1})}{(x_i-x_{i-1})(x_i-x_{i+1})} y_i \\
& +\frac{(x-x_{i-1})(x-x_i)}{(x_{i+1}-x_{i-1})(x_{i+1}-x_i)} y_{i+1}
\end{aligned}
$$

对于给定的插值点 x 选择三个插值结点的方法归纳如下:

$$
i=\begin{cases}
2 & \text{当 } x\leqslant x_2 \\
k-1 & \text{当 } x_{k-1}<x\leqslant x_k \text{ 且 } |x-x_{k-1}|\leqslant|x-x_k| \\
 & \quad (k=3, 4, \cdots, n-1) \\
k & \text{当 } x_{k-1}<x\leqslant x_k \text{ 且 } |x-x_{k-1}|>|x-x_k| \\
 & \quad (k=3, 4, \cdots, n-1) \\
n-1 & \text{当 } x>x_{n-1}
\end{cases}
$$

三、使用说明

1. 子程序语句

SUBROUTINE LAQ(N, T, X, Y, A)

2. 哑元说明

输入参数:

N　整变量, 结点个数, 且 N≥3。

T　实变量, 插值点。

X　　N个元素的一维实数组，存放给定的插值结点，要求 X(I)<X(I+1)(I=1, 2, …, N-1)。

Y　　N个元素的一维实数组，存放插值结点上相应的函数值。

输出参数:

A　　实变量，插值结果。

四、程序

```
      SUBROUTINE LAQ(N, T, X, Y, A)
      DIMENSION X(N), Y(N)
      M=N-1
      DO 10 I=3, M
      IF(T.GT.X(I)) GOTO 10
      IF(ABS(T-X(I-1)).LE.ABS(T-X(I))) I=I-1
      GOTO 20
10    CONTINUE
      I=N-1
20    U=(T-X(I))*(T-X(I+1))/(X(I-1)-X(I))/(X(I-1)-X(I+1))
      V=(T-X(I-1))*(T-X(I+1))/(X(I)-X(I-1))/(X(I)-X(I+1))
      W=(T-X(I-1))*(T-X(I))/(X(I+1)-X(I-1))/(X(I+1)-X(I))
      A=U*Y(I-1)+V*Y(I)+W*Y(I+1)
      RETURN
      END
```

五、例题

已知如表1-2所示的列表函数，求$x=0.57891$处的函数值。

表 1-2

x	0.4	0.5	0.6	0.7	0.8
y	0.38942	0.47943	0.56464	0.64422	0.71736

用数组X的元素顺序存放0.4, 0.5, 0.6, 0.7, 0.8; 数组Y的元素顺序存放0.38942, 0.47943, 0.56464, 0.64422,

0.71736。

程序如下：

```
    SUBROUTINE  LAQ(N, T, X, Y, A)
    ⋮本子程序段体部分
    END

    DIMENSION X(5), Y(5)
    READ(5, 10)X, Y
10  FORMAT(5F6.3/5F8.6)
    T=0.57891
    CALL LAQ(5, T, X, Y, A)
    WRITE(5, 20)A
20  FORMAT(20X, 'A=', F8.5)
    STOP
    END
```

计算结果：　　$y(0.57891)=0.54714$

1.3 埃特金插值

一、功能

从给定的n个插值结点中选取最靠近插值点x的相邻的$m(m\leqslant n)$个插值结点，用埃特金反复线性插值公式对一元函数进行插值。

二、方法简介

设给定的n个插值结点$x_1<x_2<\cdots<x_n$及其对应的函数值$y_1, y_2, \cdots, y_n$。从n个插值结点中选取最靠近插值点x且尽量使x位于其中心的m个结点，通过反复线性插值逐步产生一次、二次直至$m-1$次拉格朗日插值多项式，具体方法如下：

先利用线性插值公式：

$$y_1^{(k)}(x)=y_i+\frac{x-x_i}{x_{i+k}-x_i}(y_{i+k}-y_i)\quad (k=1,\ 2,\ \cdots,\ m-1) \tag{1}$$

计算 $m-1$ 个一次多项式 $y_1^{(1)}(x)$, $y_1^{(2)}(x)$, $\cdots$, $y_1^{(m-1)}(x)$。然后，再利用递推的线性插值公式：

$$y_l^{(k)}(x)=y_{l-1}^{(1)}(x)+\frac{x-x_j}{x_{j+k}-x_j}\left[y_{l-1}^{(k+1)}(x)-y_{l-1}^{(1)}(x)\right] \tag{2}$$

$$(l=2,\ 3,\ \cdots,\ m-1;\ k=1,\ 2,\ \cdots,\ m-l;\ j=i+l-1)$$

依次计算 $y_2^{(1)}(x)$, $y_2^{(2)}(x)$, $\cdots$, $y_2^{(m-2)}(x)$; $y_3^{(1)}(x)$, $y_3^{(2)}(x)$, $\cdots$, $y_3^{(m-3)}(x)$; $\cdots$, $y_{m-1}^{(1)}(x)$。则 $y_{m-1}^{(1)}(x)$就是 $m-1$ 次拉格朗日插值多项式 $y(x)$，公式(1)，(2)称为埃特金插值公式。

三、使用说明

1．子程序语句

SUBROUTINE　ATK(X, Y, T, N, M, Z, F, A)

2．哑元说明

输入参数：

N　整变量，插值结点个数。

M　整变量，选用的插值结点个数，且 M≤N。

X　N 个元素的一维实数组，存放给定的插值结点，要求 X(I)<X(I+1)(I=1, 2, …, N-1)。

Y　N 个元素的一维实数组，存放插值结点上相应的函数值。

T　实变量，插值点。

输出参数：

A　实变量，插值结果。

工作单元：

Z, F　都是 M 个元素的一维实数组。

四、程序

```
SUBROUTINE  ATK(X, Y, T, N, M, Z, F, A)
DIMENSION F(M), X(N), Y(N), Z(M)
IF (M.GT.N) M=N
```

```
      DO 1 I=1, N
      IF(T.LE.X(I)) GOTO 2
    1 CONTINUE
      I=N
    2 IF (T.NE.X(I)) GOTO 3
      F(M)=Y(I)
      GOTO 9
    3 IF (MOD(M, 2).EQ.0) GOTO 4
      IF (I.EQ.1) GOTO 4
      IF ((T-X(I-1)).GE.(X(I)-T)) GOTO 4
      I=I-1
    4 I=I-M/2
      IF(I.GT.0) GOTO 5
      I=1
      GOTO 6
    5 IF ((I+M).GT.N) I=N-M+1
    6 DO 7 J=1, M
      Z(J)=T-X(I)
      F(J)=Y(I)
    7 I=I+1
      M1=M-1
      DO 8 I=1, M1
      FI=F(I)
      ZI=Z(I)
      I1=I+1
      DO 8 J=I1, M
    8 F(J)=FI+ZI*(F(J)-FI)/(ZI-Z(J))
    9 A=F(M)
      RETURN
      END
```

五、例题

已知如表 1-3 所示的列表函数，求 $x=0.462$, 0.6 处的函数值。

表 1-3

x	0.3	0.4	0.5	0.6	0.7
y	0.29850	0.39646	0.49311	0.58813	0.68122

用数组 X 的元素顺序存放0.3, 0.4, 0.5, 0.6, 0.7; 数组 Y 的元素顺序存放 0.29850, 0.39646, 0.49311, 0.58813, 0.68122; 取 M=4。

程序如下:

```
    SUBROUTINE  ATK(X, Y, T, N, M, Z, F, A)
    :本子程序段体部分
    END

    DIMENSION X(5), Y(5), T(2), Z(4), F(4)
    READ (5, 5) X, Y
5   FORMAT (5F5.2/5F8.6)
    T(1)=0.462
    T(2)=0.6
    DO 10 I=1, 2
    CALL ATK(X, Y, T(I), 5, 4, Z, F, A)
10  WRITE (4, 20) T(I), A
20  FORMAT (6X, 2HX=, F6.3, 5X, 2HY=, F8.5)
    STOP
    END
```

计算结果:

$$y(0.462)=0.45656, \quad y(0.6)=0.58813$$

1.4 三次自然样条函数插值、微商与积分

一、功能

给定 n 个插值结点 $x_1<x_2<\cdots<x_n$ 及对应的函数值 $y_1, y_2, \cdots, y_n$, 用三次自然样条函数 $S(x)$, 求一组插值点上的函数值、一阶导数值、二阶导数值和定积分 $\int_{x_1}^{x_n} S(x)dx$ 的值。

二、方法简介

三次自然样条函数 $S(x)$ 在每个分段 (x_{i-1}, x_i) 上都是三次多项式，且满足下列条件：

1．插值条件：$S(x_i)=y_i(i=1, 2, \cdots, n)$

2．连接条件：在分点 x_i 处具有连续的一阶和二阶导数，即

$$\left.\begin{aligned} S'(x_i-0)=S'(x_i+0) \\ S''(x_i-0)=S''(x_i+0) \end{aligned}\right\} \quad (i=2, 3, \cdots, n-1)$$

3．自然边界条件：$S''(x_1)=S''(x_n)=0$

在每个子区间 $[x_i, x_{i+1}]$ 上，由牛顿插值公式有

$$\begin{aligned} S(x)=&S(x_i)+(x-x_i)S(x_i, x_{i+1}) \\ &+(x-x_i)(x-x_{i+1})S(x, x_i, x_{i+1}) \end{aligned} \tag{1}$$

这里
$$S(x_i, x_{i+1})=\frac{S(x_{i+1})-S(x_i)}{x_{i+1}-x_i}$$

$$S(x, x_i, x_{i+1})=\frac{1}{6}[S''(x)+S''(x_i)+S''(x_{i+1})]$$

因 $S(x)$ 是三次多项式，$S''(x)$ 在 $[x_i, x_{i+1}]$ 上是一次多项式，故

$$S''(x)=S''(x_i)+(x-x_i)S''(x_i, x_{i+1}) \tag{2}$$

对(1)求导，则有

$$\begin{aligned} S'(x)=&S(x_i, x_{i+1})+(2x-x_i-x_{i+1})S(x, x_i, x_{i+1}) \\ &+\frac{1}{6}(x-x_i)(x-x_{i+1})S''(x_i, x_{i+1}) \end{aligned} \tag{3}$$

利用连接条件，由此可得 $S''(x_i)(i=2, 3, \cdots, n-1)$ 满足三对角线性方程组

$$\begin{aligned} &(x_i-x_{i-1})S''(x_{i-1})+2(x_{i+1}-x_{i-1}) \\ &\cdot S''(x_i)+(x_{i+1}-x_i)S''(x_{i+1}) \\ &=6[S(x_i, x_{i+1})-S(x_{i-1}, x_i)] \end{aligned} \tag{4}$$

$$(i=2, 3, \cdots, n-1)$$